

## Software-Anleitung

# Das Excel-Werkzeug "Tabelle" im Praxiseinsatz

## Teil 1: Mit strukturierten Verweisen die Datenanalyse vereinfachen

Wer Daten in Excel verwaltet, arbeitet in der Praxis oft auch mit größeren Listen. Die Pflege solcher Listen wird allerdings mit zunehmender Größe und Komplexität der Daten immer schwieriger. Denn anders als in Datenbanken passen sich in Excel die Zellverweise in den Auswertungsformeln nicht automatisch an, wenn sich die Größe oder Form der Liste ändert. Dadurch ist es erforderlich, Argumente und Zellverweisen in Formeln, PivotTable-Bereichen und Diagrammen zu kontrollieren, was bei umfangreichen Listen viel Zeit kostet und ein hohes Fehlerrisiko birgt – denn Excel ermöglicht ab Version 2007 immerhin Tabellenblätter mit bis zu 1.048.576 Zeilen und 16.384 Spalten.

Um auch umfangreiche Listen gut handhaben zu können, stellt Excel ab Version 2007 ein neues Werkzeug zur Verfügung: Listen können mit wenigen Klicks in "Tabellen" umgewandelt werden, für die neue und wesentlich einfachere Kalkulations- und Analysetechniken zur Verfügung stehen. Dieser Beitrag erklärt die Funktionsweise von Tabellen und zeigt, welche Vorteile diese im praktischen Einsatz haben.

**!** Die Bezeichnung "Tabelle" ist etwas irreführend, da Excel-Anwender unter dem Begriff "Tabelle" das ganze Tabellenblatt verstehen. Wer zukünftig mit der neuen Listentechnik arbeitet, muss deshalb die Begriffe "Tabelle" und "Tabelleblatt" klar voneinander trennen.

**Autor****Ignatz Schels**

Seit 1986 selbst. DV-Dozent und Journalist, Leitet u. konzipiert MS-

Project-Seminare u.a. für DaimlerChrysler, Siemens, T-Systems

Kontakt: [info@schels.de](mailto:info@schels.de)

Mehr Informationen unter: [projektmagazin.de/autoren](http://projektmagazin.de/autoren)

**ähnliche Artikel****in der Rubrik:**

› [Microsoft Excel](#)

**Service-Links****Bücher**

› [Microsoft Excel](#)

=SVERWEIS(B2;Lieferanten!\$A\$1:\$F\$67432;2;FALSCH)				
=SVERWEIS(A2;Projekte!\$A\$1:\$B\$256453;2;FALSCH)				
	A	B	C	D
1	Kontierung	LieferantenNr	Sverweis Projekte	Sverweis Lieferanten
2	5921061391	10100659	Actros Getriebedesign	Designcenter Stuttgart
3	5921061391	10100660	Actros Getriebedesign	Werk Köngen
4	5921061391	10100661	Actros Getriebedesign	Achters GmbH
5	5921061629	10100662	VAN Neukonstruktion	Alzmann
6	5921061629	10100663	VAN Neukonstruktion	Bose GmbH Bamberg
7	5921061629	10100664	VAN Neukonstruktion	Rectinel
8	5921063507	10100665	Neubau Maschinenhalle Düren	Alcatron
9	5921063507	10100666	Neubau Maschinenhalle Düren	Johnson
10	5921063507	10100667	Neubau Maschinenhalle Düren	Bickler Räder & Rollen

Bild 1: Unübersichtlich und schwer zu kontrollieren: Zellverweise auf große externe Listen, wie hier am Beispiele eines SAP-Projektberichts.

## Vorteile von Tabellen

"Tabellen" bieten gegenüber herkömmlichen Listen einige wesentliche Vorteile:

- Tabellen **erweitern sich automatisch**, wenn neue Daten angefügt werden. Gibt man z.B. Daten in die leere Zeile unterhalb der Tabelle ein, kopiert Excel automatisch alle Formeln aus der vorletzten Zeile in diese Zeile.
- Tabellen können in der **Ergebniszeile** am unteren Rand einzelner Spalten automatisch die Summe, die Anzahl von Daten, die Mittelwerte etc. anzeigen. Die Ergebnisse sind auch dann korrekt, wenn die Filterfunktion verwendet wird, da sich die Formel TEILERGEBNIS() in der Ergebniszeile automatisch nur auf die Daten des Filterergebnisses bezieht.
- Zur Kalkulation dienen in Tabellen **strukturierte Verweise**. Diese ermöglichen ein intuitiveres Arbeiten als Zellbezüge, da sie nicht auf einzelne Zellen oder Zellbereiche verweisen (z.B. "\$A\$1"), sondern auf Elemente der Tabellenstruktur, wie *Zeile*, *Spalte* oder *Daten*. (z.B. "[SpalteA]").

## Von der Liste zur Tabelle

Bevor Sie diese Vorteile nutzen können, müssen Sie Ihre Liste zunächst in eine "Tabelle" umwandeln. Wie das geht, zeigt dieses erste Beispiel, das Sie in der Beispieldatei im Tabellenblatt "Projekte2013" finden. Die Liste in Bild 2 beginnt mit einer Kopfzeile, die Daten entsprechen dem Datentyp der Spalte – d.h. die Datumsspalten enthalten nur Datumswerte, in der Zahlenspalte befinden sich nur numerische Werte.

- Klicken Sie in eine beliebige Zelle der Liste.
- Wählen Sie *Einfügen / Tabellen / Tabelle* und öffnen Sie damit das Dialogfenster "Tabelle erstellen". Excel schlägt den Bereich rund um die markierte Zelle als Bezug vor und erkennt die Überschrift der Liste automatisch. Sie können den Bereich bei Bedarf noch korrigieren, indem Sie im Hintergrund den Mauszeiger über den gewünschten Listenbereich ziehen.
- Bestätigen Sie mit OK, damit die Liste in eine Tabelle umgewandelt wird.
- Lassen Sie den Zellzeiger in der Tabelle und weisen Sie dieser eine Formatvorlage zu, indem Sie rechts im Menüband die Registerkarte *Entwurf* in der Gruppe *Tabellentools* wählen und die gewünschte Vorlage anklicken (Bild 4). Um die Auswahl zu erleichtern, zeigt Excel eine Vorschau an, sobald Sie mit dem Mauszeiger über eine der angebotenen Varianten fahren.

	A	B	C	D
1	Projekt	Beginn	Ende	Budget
2	Karosseriebau Instandhaltung	01.10.2013	29.01.2014	120.000
3	Verbesserung Elektronik	01.10.2013	03.02.2014	150.000
4	CDK Produktpflege	15.10.2013	22.02.2014	125.000
5	Ausarbeitung Richtlinien Umweltschutz	15.10.2013	22.02.2014	160.000
6	Getriebestrang Automatisierung	01.11.2013	31.03.2014	90.500
7	Kapazitätserhebung und Pers.planung	01.11.2013	31.03.2014	135.000
8	Unfallverhütungsvorschriften	15.11.2013	15.03.2014	56.000
9	Entwicklung e-commerce-Lösung	15.11.2013	15.03.2014	260.500
10	CDLK Serie II	01.12.2013	30.04.2014	490.000
11	ABM-V Serie VI	15.12.2013	14.05.2014	320.000

Bild 2: Projektkliste




Bild 3: Excel analysiert die Liste und schlägt den Datenbereich automatisch vor.

	A	B	C	D	E	F
1	Projekt	Beginn	Ende	Budget		
2	Karosseriebau Instandhaltung	01.10.2013	29.01.2014	120.000		
3	Verbesserung Elektronik	01.10.2013	03.02.2014	150.000		
4	CDK Produktpflege	15.10.2013	22.02.2014	125.000		
5	Ausarbeitung Richtlinien Umweltschutz	15.10.2013	22.02.2014	160.000		
6	Getriebestrang Automatisierung	01.11.2013	31.03.2014	90.500		
7	Kapazitätserhebung und Pers.planung	01.11.2013	31.03.2014	135.000		
8	Unfallverhütungsvorschriften	15.11.2013	15.03.2014	56.000		
9	Entwicklung e-commerce-Lösung	15.11.2013	15.03.2014	260.500		
10	CDLK Serie II	01.12.2013	30.04.2014	490.000		
11	ABM-V Serie VI	15.12.2013	14.05.2014	320.000		

Bild 4: Die neue Tabelle und Tabellenformatvorlagen

Die Tabelle ist jetzt fertig erstellt. Neue Einträge schreiben Sie einfach in die nächste freie Zeile. Mit jedem Eintrag erweitert sich die Tabelle automatisch um eine Zeile. Das Tabellenformat passt sich dabei an, die Zelle wird entsprechend eingefärbt. Enthalten einzelne Spalten der Tabelle Formeln, werden diese automatisch auf die neu angefügte Zeile kopiert.

! Am rechten unteren Tabellenende sehen Sie ein kleines blaues Dreieck. Wenn Sie dieses mit gedrückter Maustaste ziehen, können Sie die Tabellengröße ändern, z.B. um Leerzeilen anzufügen. Alternativ dazu können Sie den Bereich jederzeit nachbessern, indem Sie die Option *Tabellengröße ändern* unter *Tabellentools / Entwurf / Eigenschaften* wählen.

## Der Tabellename

Mit der Umwandlung des Listenbereichs in eine Tabelle hat diese einen Namen bekommen. Wenn Sie auf *Tabellentools / Entwurf* klicken, sehen Sie diesen Namen in der Gruppe *Eigenschaften* links oben. Er passt sich automatisch an die Tabellendimension an, was für spätere Auswertungen besonders hilfreich ist. Sie können den Tabellennamen jederzeit ändern. Für die Namensgebung in Tabellen gelten die gleichen Richtlinien wie für Bereichsnamen: Das erste Zeichen muss ein Buchstabe sein, Zellbezüge und Leerzeichen sind nicht erlaubt, die maximale Länge beträgt 255 Zeichen. Verwenden Sie am besten nach den Namenskonventionen für Tabellen den Vorsatz "tbl", damit Sie Tabellennamen von Bereichsnamen unterscheiden können:

"tblProjekte"

! Der Tabellename bezieht sich auf den Datenbereich der Tabelle, schließt also die Kopfzeile und die Ergebniszeile nicht mit ein. Das unterscheidet Tabellennamen von Bereichsnamen, die in der Regel auf ganze Listen inklusive Kopfzeile verweisen.

Im *Namens-Manager* (unter *Formeln / Definierte Namen*) finden Sie eine Übersicht über alle Bereichsnamen sowie über alle Tabellennamen in der aktiven Arbeitsmappe. Während Bereichsnamen wahlweise entweder in der gesamten Mappe verfügbar sind (globale Bereichsnamen) oder nur im aktuellen Tabellenblatt (lokale Bereichsnamen), sind Tabellennamen immer global, d.h. Sie können diese in allen Tabellenblättern der Mappe verwenden. Der Bezug, auf den ein Tabellenname verweist, lässt sich im Namens-Manager nicht ändern – dies ist nur nach einem Klick in die Tabelle im Register *Tabellentools* unter *Eigenschaften* möglich.

### Von der Tabelle zurück zur Liste


Sie können eine Tabelle problemlos wieder in eine Liste zurückverwandeln, z.B. wenn Sie die Tabelle an jemanden weitergeben wollen, der mit einer Excel-Version älter als 2007 arbeitet, die das Tabellenformat nicht kennt. Die Option dafür finden Sie unter *Tabellentools / Entwurf / Tools*. Mit *In Bereich konvertieren* werden alle tabellenspezifischen Elemente entfernt. Die Ergebniszeile mit den TEILERGEBNIS()-Funktionen bleibt allerdings erhalten.

Wollen Sie auch die Tabellenformatvorlage abschalten, sodass die Liste wieder farblos und unformatiert ist, markieren Sie den bisherigen Tabellenbereich, wählen Sie *Start / Formatvorlagen / Zellenformatvorlagen* und klicken Sie auf *Standard*.

### Die Ergebniszeile

Wenn Sie unter *Tabellentools / Entwurf / Optionen für Tabellenformat* die Ergebniszeile einschalten, steht Ihnen in der Abschlusszeile jeder einzelnen Spalte eine Auswahlliste mit Funktionen bereit, um z.B. die Spaltenwerte zu zählen, zu summieren oder mit statistischen Funktionen wie Mittelwert, Maximum, Minimum und Standardabweichung auszuwerten. Excel verwendet dazu die Funktion TEILERGEBNIS(), die mit Parametern arbeitet, um die Berechnungsart festzulegen. So steht z.B. der Parameter "109" für die Berechnung der Summe.

**!** Ist die Ergebniszeile aktiv, funktioniert die automatische Erweiterung der Tabelle bei neuen Einträgen nicht mehr. Fügen Sie in diesem Fall einfach Leerzeilen vor der Ergebniszeile ein oder erweitern Sie die Tabelle mit dem Dreieckssymbol rechts unten.



	A	B	C	D	E
1	Projekt	Beginn	Ende	Budget	
2	Karosseriebau Instandhaltung	01.10.2013	29.01.2014	120.000	
3	Verbesserung Elektronik	01.10.2013	03.02.2014	150.000	
4	CDK Produktpflege	15.10.2013	22.02.2014	125.000	
5	Ausarbeitung Richtlinien Umweltschutz	15.10.2013	22.02.2014	160.000	
6	Getriebestrang Automatisierung	01.11.2013	31.03.2014	90.500	
7	Kapazitätserhebung und Pers.planung	01.11.2013	31.03.2014	135.000	
8	Unfallverhütungsvorschriften	15.11.2013	15.03.2014	56.000	
9	Entwicklung e-commerce-Lösung	15.11.2013	15.03.2014	260.500	
10	CDLK Serie II	01.12.2013	30.04.2014	490.000	
11	ABM-V Serie VI	15.12.2013	14.05.2014	320.000	
12	Ergebnis	10		1.907.000	
13				Ohne	
14				Mittelwert	
15				Anzahl	
16				Anzahl Zahlen	
17				Maximum	
18				Minimum	
19				Summe	
20				Standardabweichung	
21				Varianz (Stichprobe)	
22				Weitere Funktionen...	

Bild 5: Ergebniszeile mit Funktionen für die Spaltenauswertung

## Strukturierte Verweise

Die in Excel ab Version 2007 neu eingeführten strukturierten Verweise vereinfachen die Konstruktion von Formeln erheblich, da die Formeln jetzt mit Elementen der Tabelle arbeiten, z.B. mit dem Spaltennamen oder mit dem

Zeilenwert einer Spalte, die neben der Formelzelle liegt. Das gleiche gilt für Formeln, die "von außen" auf eine Tabelle oder Spalten einer Tabelle verweisen.

Das Prinzip der strukturierten Verweise lässt sich am einfachsten an einem Beispiel verdeutlichen: In der Beispielliste aus Bild 2 soll die Gesamtsumme der Budgetspalte mit Hilfe der Funktion TEILERGEBNIS() berechnet werden (die Funktion liefert auch beim Einsatz eines Filters noch korrekte Ergebnisse). Der Formelparameter "109" steht dabei für "SUMME()", die Formel bezieht sich auf den Zellbereich \$D\$2:\$D\$11. Die Formel lautet also:

=TEILERGEBNIS(109;\$D\$2:\$D\$11)

Haben Sie die Liste jedoch in eine Tabelle umgewandelt, können Sie den absoluten Bezug einfach durch den Spaltennamen ersetzen. Die Formel lautet dann:

=TEILERGEBNIS(109;[Budget])

Kommen neue Zeilen hinzu, passt die Formel automatisch ihr Ergebnis an, was mit dem absoluten Bezug nicht möglich ist.

### Bezeichner

Für einen strukturierten Verweis auf eine Spalte genügt die Angabe des in eckige Klammern gesetzten Spaltennamens. Neben diesem einfachen gibt es noch einige besondere "Bezeichner", die Tabelle 1 aufführt.

Bezeichner	Bedeutung
[#Alle]	Bezug auf die gesamte Tabelle mit Spaltennamen, Daten und Ergebniszeile
[#Daten]	Bezug auf die Daten ohne die Ergebniszeile
[#Kopfzeilen]	Bezug auf die Kopfzeile mit den Spaltennamen
[#Ergebnisse]	Bezug auf die Ergebniszeile, ergibt 0, wenn diese nicht vorhanden ist
[@]	Bezug auf die Daten in dieser Zeile

Tabelle 1: Bezeichner in strukturierten Verweisen.

Der Bezeichner **[#Alle]** wird für Auswertungen selten zum Einsatz kommen, da er die Tabelle inklusive Kopfzeile und Ergebniszeile einschließt. Diese Formeln berechnen z.B. die Größe der Tabelle:

=ZEILEN(tblProjekte[#Alle]) und =SPALTEN(tblProjekte[#Alle])

Der Bezeichner **[#Daten]** adressiert in Formeln den Datenbereich der Tabelle ohne Kopfzeile und Ergebniszeile, kann aber in der Praxis meist weggelassen werden. Hier wird mit der Funktion ZEILE() die Größe des Datenbereichs berechnet:

=ZEILEN(tblProjekte[#Daten]) oder =ZEILEN(tblProjekte)

Die nachfolgende Formel überprüft, ob die Ergebniszeile der Tabelle eingeschaltet ist, indem sie die Zeilenzahl der Tabelle mit der Zeilenzahl des Datenbereiches vergleicht:

=WENN(ZEILEN(tblProjekte[#Alle])-1>ZEILEN(tblProjekte[#Daten]));"enthält Ergebniszeile";"keine Ergebniszeile")

Sehr nützlich für Verweise auf Tabellen ist der Bezeichner **[#Kopfzeilen]**. Er enthält die erste Zeile der Tabelle und kann als Matrix in Formeln verwendet werden. Die folgende Formel sucht beispielsweise nach einem Spaltennamen in der Tabelle. Das Ergebnis ist die Spaltennummer oder ein #WERT-Fehler, falls die Spalte nicht existiert:

=VERGLEICH("Projekt";tblProjekte[#Kopfzeilen];0)

Der Bezeichner **[#Ergebnisse]** lässt sich ebenfalls als Matrix auswerten. Ist die Spaltennummer bekannt, liefert die Funktion INDEX() den Wert aus der Ergebniszeile:

=INDEX(tblProjekte[#Ergebnisse];1;4)

Mit dem Bezeichner **[@]** holt die Formel denjenigen Wert aus der Spalte, der sich parallel zur Formelzelle befindet (d.h. in derselben Zeile wie die Formel steht). **[@]** erfüllt dabei die Funktion, die früher der relative Bezug hatte.

=[@Budget]      Der Wert aus der Spalte *Budget*, relativ zur Formelzelle

=[Budget]      Der absolute Wert der Spalte *Budget*

## Operatoren

Strukturierte Verweise verwenden neben den allgemeinen arithmetischen Rechenoperatoren ( +, -, \*, / ) noch folgende zusätzliche Operatoren:

Operator	Beispiel	Entspricht Zellbezug
Doppelpunkt (:)	=Tabelle1[Nord]:[Ost]	A2:D7
Komma (,)	=Tabelle1[Nord], Tabelle1[Süd]	A2:A7,B2:B7
Schnittmenge (Leertaste)	=Tabelle1[Nord] Tabelle1[Juni]	A2:A7 A6:D6

Tabelle 2: Arithmetische Operatoren in strukturierten Verweisen

## Beispiel: Projektdauer berechnen

In unserer Beispieldatei können Sie mit Hilfe eines strukturierten Verweises die Dauer der einzelnen Projekte in einer neuen Spalte berechnen. Öffnen Sie dazu in der Beispieldatei das Tabellenblatt "Projektdauer":

- Markieren Sie mit der rechten Maustaste die Spalte D und wählen Sie im Kontextmenü den Befehl *Zellen einfügen*.
- Tragen Sie in D1 den Spaltennamen *Dauer* ein.
- Zur Berechnung der Differenz zwischen Beginn und Ende konstruieren Sie in D2 die Formel  
=[@Ende]-[@Beginn]+1



Schreiben Sie zunächst das Gleichheitszeichen ("="), klicken Sie mit dem Mauszeiger auf den Wert in der Spalte Ende, schreiben Sie dann das Minuszeichen und markieren Sie den Wert aus der Spalte Beginn. Das @-Zeichen, das Excel automatisch einfügt, stellt sicher, dass der parallele Wert der Spalten verwendet wird.

	A	B	C	D	E
	Projekt	Beginn	Ende	Dauer	Budget
2	Karosseriebau Instandhaltung	01.10.2013	29.01.2014	121 Tage	120.000
3	Verbesserung Elektronik	01.10.2013	03.02.2014	126 Tage	150.000
4	CDK Produktpflege	15.10.2013	22.02.2014	131 Tage	125.000
5	Ausarbeitung Richtlinien Umweltschutz	15.10.2013	22.02.2014	131 Tage	160.000
6	Getriebestrang Automatisierung	01.11.2013	31.03.2014	151 Tage	90.500
7	Kapazitätserhebung und Pers.planung	01.11.2013	31.03.2014	151 Tage	135.000
8	Unfallverhütungsvorschriften	15.11.2013	15.03.2014	121 Tage	56.000
9	Entwicklung e-commerce-Lösung	15.11.2013	15.03.2014	121 Tage	260.500
10	CDLK Serie II	01.12.2013	30.04.2014	151 Tage	490.000
11	ABM-V Serie VI	15.12.2013	14.05.2014	151 Tage	320.000
12		10		136 Tage	1.907.000

Bild 6: Die Dauer wird über einen strukturierten Verweis berechnet.

- Bestätigen Sie die Formel in der ersten Zeile mit der Eingabe-Taste. Sie wird automatisch auf die gesamte Spalte verteilt.
- Weisen Sie der Spalte folgendes benutzerdefiniertes Zahlenformat zu, das die Differenz in Tagen ausweist: 0" Tage"
- Schalten Sie zur Berechnung der Durchschnittsdauer in der Ergebniszeile auf die Funktion *Mittelwert* um.




Bild 7: Strukturierte Verweise lassen sich durch die Eingabehilfen bequem eintippen.

## Externe Verweise

Für die Auswertung der Tabellendaten in externen Bereichen muss der Tabellennamen angegeben werden. Excel unterstützt Sie bei diesen Aktionen durch die Autovervollständigen-Funktion sowie mit Eingabehilfen. Schreiben Sie beispielsweise in der Formel nur die ersten Buchstaben des Tabellennamens, schlägt Excel Ihnen sofort den vollständigen Namen vor, den Sie mit der *Tab*-Taste übernehmen können. Sobald Sie nach dem Tabellennamen eine eckige Klammer eingeben, zeigt die Zellzeigerinfo alle verfügbaren Bezeichner an.

### Beispiel: Gesamtbudget und Durchschnittsdauer berechnen

Berechnen Sie in einer Zelle außerhalb des Tabellenbereichs die Summe aller Einzelbudgets und die Durchschnittsdauer der einzelnen Projekte. Die Ergebnisse sollten mit den Werten in der Ergebniszeile übereinstimmen.

- Schreiben Sie in eine beliebige Zelle außerhalb des Tabellenbereichs die Formel =SUMME(
- Wählen Sie mit dem Mauszeiger die Werte in der Spalte *Budget* aus (ohne Kopf- und Ergebniszeile)
- Schließen Sie die Formel durch Eingabe der Klammer ab =SUMME(tblProjekte[Budget])
- Berechnen Sie auf die gleiche Weise auch die Durchschnittsdauer =MITTELWERT(tblProjekte[Dauer])

Möchten Sie zusätzlich wissen, wie viele Projekte z.B. im Oktober beginnen, berechnen Sie dies mit folgender Formel:

=SUMME(WENN(MONAT(tblProjekte[Beginn])=10;1;0))

Da es sich um eine Matrixformel handelt, müssen Sie zum Abschluss die Tastenkombination **Strg + Umschalt + Eingabe** drücken. (Mehr zum Thema Matrixformeln, siehe **"Matrixformeln als nützliche Helfer in Microsoft Excel"**, Projekt Magazin, 19/2008)

### Beispiel: Laufende IST-Kosten berechnen

Wenn Sie eine vorhandene Liste in eine Tabelle umwandeln, sollten Sie grundsätzlich alle vorhandenen Standardformeln durch strukturierte Verweise ersetzen – vor allem wenn Sie Zahlenreihen kumulieren. Denn mit der Umwandlung von Listen in Tabellen rechnen manche Standardformeln nicht mehr richtig, wie nachfolgendes Beispiel zeigt.

Bild 8 zeigt eine Beispielliste mit monatlichen Ist-Kosten (die Liste finden Sie im Tabellenblatt "IST-Kosten" der Beispieldatei). Die laufenden Kosten lassen sich mit einer Standardformel über die Kombination aus absolutem und relativem Bezug auf den Zeilenwert ermitteln.

Wandeln Sie die Liste in eine Tabelle um und berechnen Sie in einer neuen Spalte die kumulierten Kosten mit Hilfe eines strukturierten Verweises. Die Funktion INDEX() benötigen Sie, um auf den ersten Wert der Kosten-spalte zu referenzieren (entspricht dem absoluten Bezug in der obigen Standardformel). Das @-Zeichen sorgt für den relativen Bezug auf den Zeilenwert:

`=SUMME(INDEX([Istkosten];1):[@Istkosten])`

Wenn Sie jetzt die Tabelle erweitern und den nächsten Monatsnamen sowie die Kosten eintragen, werden beide Formeln automatisch in die neue Zeile eingetragen. Die "alte" Version rechnet aber im Unterschied zum strukturierten Verweis nicht mehr richtig (Bild 9).

## Fazit

Tabellen und strukturierte Verweise sind richtungsweisende Werkzeuge für die Tabellenkalkulation mit Excel, die das Listenprinzip mit relativen und absoluten Bezügen allmählich ablösen werden. Tabellen machen die Bearbei-

	A	B	C	D
1	Monat	Istkosten	Istkosten kumuliert	
2	Januar	12.000	12.000	=SUMME(\$B\$2:B2)
3	Februar	15.000	27.000	=SUMME(\$B\$2:B3)
4	März	16.000	43.000	=SUMME(\$B\$2:B4)
5	April	13.000	56.000	=SUMME(\$B\$2:B5)
6	Mai	15.000	71.000	=SUMME(\$B\$2:B6)
7	Juni	21.000	92.000	=SUMME(\$B\$2:B7)

Bild 8: Die Standardformel für kumulierte Werte in Listen

	A	B	C	D	E
1	Monat	Istkosten	Istkosten kumuliert	laufende Kosten	
2	Januar	12.000	12.000	12.000	
3	Februar	15.000	27.000	27.000	
4	März	16.000	43.000	43.000	
5	April	13.000	56.000	56.000	
6	Mai	15.000	71.000	71.000	
7	Juni	21.000	92.000	92.000	
8	Juli	23.000	115.000	115.000	

Bild 9: Standardformeln rechnen in Tabellen nicht mehr richtig.



tung großer Datenmengen komfortabler und sicherer und geben dem Anwender mehr Möglichkeiten für die Analyse seiner Daten. Für viele Excel-Anwender bedeutet das zwar, langjährige Gewohnheiten aufzugeben, "Trampelpfade" zu verlassen und neue Wege zu gehen, aber es wird sich lohnen.

## Ausblick: Tabellen und ODBC

Excel stellt mit ODBC (open database connectivity) seit Version 95 ein Werkzeug für den Import externer Daten bereit. ODBC ermöglicht es, Tabellen und Abfragen aus anderen Excel-Mappen, aber auch aus Datenbanken wie Access, Oracle, SQL-Server u.a. in Excel zu integrieren, wobei die Daten dynamisch mit der Datenquelle verknüpft werden. Seit Version 2007 ist die hier beschriebene Technik der Tabellen und strukturierten Verweise in die ODBC-Verknüpfung einbezogen.

Der zweite und abschließende Teil dieses Beitrags beschreibt ausführlich, wie Sie Tabellen und strukturierte Verweise in Kombination mit ODBC-Verknüpfungen nutzen. Am Beispiel einer Mehrprojektverwaltung lernen Sie den Aufbau von ODBC-Verbindungen, die Weiterverarbeitung importierter Daten mit strukturierten Verweisen und die Konsolidierung von Tabellen mit der Datenbank-Abfragesprache SQL kennen.




Bild 10: Mit ODBC importierte Daten werden automatisch in Tabellen umgewandelt

Hat Ihnen dieser Artikel gefallen?

Bewerten Sie ihn im Projekt Magazin online und teilen Sie so Ihre Meinung anderen Lesern mit. Wählen Sie dazu den Artikel im Internet unter [www.projektmagazin.de/ausgaben/2013](http://www.projektmagazin.de/ausgaben/2013) oder klicken Sie [hier](#), um direkt zum Artikel zu gelangen.